MP2

Le but de ce mini-projet est d'utiliser le solveur SMT Z3 pour résoudre des instances du problème de théorie des graphes suivant, pour $s, d, c \in \mathbb{N}$ donnés:

Existe-t-il un graphe simple et non dirigé, ayant s sommets tous de degrè au moins d, n'ayant aucune clique de taille c?

Voici une manière d'exprimer ce problème en SMT-LIB, dans le cas $s = 4, d = 2, c = 3$:

```
(declare-datatypes () ((S s4 s3 s2 s1)))
; l'ensemble des sommets du graphe est S={s1,s2,s3,s4}

(declare-fun A (S S) Bool)
; le predicat A(si,sj) est vrai si l'arete entre les sommets
; si et sj existe, faux sinon

(assert (forall ((x S)(y S)) (=> (A x y) (not(= x y)))))
; pas d'aretes qui bouclent

(assert (forall ((x S)(y S)) (=> (A x y) (A y x))))
; le graphe n'est pas dirige

(assert  (forall ((x S)) (exists ((x2 S)(x1 S))
                     (and (A x x2)(A x x1)(not(= x2 x1))))))
; le degre de chaque sommet est au moins 2

(assert  (forall ((x3 S)(x2 S)(x1 S))
         (or (not (A x3 x2))(not (A x3 x1))(not (A x2 x1)))))
; il n'existe aucune clique de taille 3.
; On remarque qu'il n'est pas necessaire de demander
; que les sommets x1, x2, x3 soient distincts car, si
; un meme sommet est pris deux fois, l'absence d'aretes qui bouclent
; implique que la formule "(or (not (A x3 x2))..." est verifiee.

(check-sat)

(get-model)
; si la conjonction des assertions ci-dessus a un modele, on recupere
; une interpretation de A qui les satisfait toutes. Cette interpretation
; de A  decrit un graphe a 4 sommets de degre au moins 2 sans 3-cliques
```

Cette instance du problème admet une solution, fournie comme suit par Z3:

```
sat
(model
  (define-fun A ((x!0 S) (x!1 S)) Bool
    (or (and (= x!0 s3) (= x!1 s2))
        (and (= x!0 s4) (= x!1 s1))
        (and (= x!0 s1) (= x!1 s2))
        (and (= x!0 s2) (= x!1 s1))
        (and (= x!0 s1) (= x!1 s4))
        (and (= x!0 s2) (= x!1 s3))
        (and (= x!0 s3) (= x!1 s4))
        (and (= x!0 s4) (= x!1 s3))))
)
```

Le graphe décrit par cette interprétation du prédicat A est

$$G = (\{s1, s2, s3, s4\}, \{\{s1, s2\}, \{s1, s4\}, \{s2, s3\}, \{s3, s4\}\})$$

En effet, les quatre sommets de G ont degrè deux, et il n'existe aucune clique de taille trois dans G.

Exercice 1 *1. En vous inspirant du cas $s = 4, d = 2, c = 3$ décrit ci-dessus, écrivez un fichier source SMT-LIB pour le cas $s = 4, d = 3, c = 3$, et vérifiez que la sortie de Z3 est dans ce cas*

```
unsat
(error "...: model is not available")
```

2. Déterminez, en utilisant Z3, le plus petit $n \in \mathbb{N}$ tel que l'instance $s = n, d = 3, c = 3$ a une solution. Pour cela, vous écrirez les fichiers source des cas $s = 5, d = 3, c = 3$, $s = 6, d = 3, c = 3$,...jusqu'à trouver une instance résoluble.

Exercice 2 *Écrivez une fonction OCaml* `generate_instance:unit -> unit` *qui demande à l'utilisateur les valeurs de s, d et c et qui écrit dans un fichier* `output.z3` *le code source SMT-LIB pour vérifier l'existence d'un graphe ayant s sommets tous de degrè au moins d sans cliques de taille c.*

Utilisez cette fonction pour explorer les limites de la méthode, sur votre ordinateur. Par exemple:

- *vérifiez qu'il existe un graphe ayant 10 sommets tous de degrè au moins 5 sans 3-cliques. Temps d'exécution?*
- *Arrive-t-on a vérifier si un graphe ayant 10 sommets tous de degrè au moins 6 peut ne pas avoir de 4-cliques?*
- *...*

Décrivez vos expériences dans le ficher RENDU.